Machin's Series Expansion for π

Helmut Knaust, Department of Mathematical Sciences, UTEP, El Paso TX 79968
hknaust@utep.edu
8/8/09

Gregory's Approximation

James Gregory (1638 - 1675) discovered the Taylor series expansion with center 0 for the arctan function:

```
Series[ArcTan[x], {x, 0, 21}]
```

$$x-\frac{x^3}{3}+\frac{x^5}{5}-\frac{x^7}{7}+\frac{x^9}{9}-\frac{x^{11}}{11}+\frac{x^{13}}{13}-\frac{x^{15}}{15}+\frac{x^{17}}{17}-\frac{x^{19}}{19}+\frac{x^{21}}{21}+O(x^{22})$$

One can use this series to approximate π. Note that π = 4 arctan (1).

```
ArcTan[1]
4 ArcTan[1.]
```

$$\frac{\pi}{4}$$

3.14159

Convergence is very slow. Below is the approximation of π using the first 1,000 terms of the series.

```
4 Normal[Series[ArcTan[x], {x, 0, 2000}]] /. {x → 1.}
```

3.14059

Machin's Formula

John Machin (1680 - 1752) used trigonometric identities to develop a much more efficient formula for approximating π:

$$\frac{\pi}{4}=4\arctan\left(\frac{1}{5}\right)-\arctan\left(\frac{1}{239}\right)$$

Machin used this formula in 1706 to compute the first 100 digits of π.
Below is Machin's approximation using the first 10 terms of the Taylor series for the arctan function:

```
f[x_] = Normal[Series[ArcTan[x], {x, 0, 20}]]
```

$$-\frac{x^{19}}{19}+\frac{x^{17}}{17}-\frac{x^{15}}{15}+\frac{x^{13}}{13}-\frac{x^{11}}{11}+\frac{x^9}{9}-\frac{x^7}{7}+\frac{x^5}{5}-\frac{x^3}{3}+x$$

```
N[π, 30]
4 N[4 f[1 / 5] - f[1 / 239], 30]
```

3.14159265358979323846264338328

3.14159265358979169691727961962

The difference between the approximation and π is smaller than 2×10^{-15}:

```
Abs[N[π, 30] - 4 N[4 f[1 / 5] - f[1 / 239], 30]]
```

$1.54154536376366\times10^{-15}$

Problem 1:
How many terms of the series are needed to compute the first 100 digits of π?

Our estimation of the accuracy above depended on *Mathematica*'s ability to compute π accurately to as many digits as desired. Alternatively, one can use an error estimate for the Taylor polynomial.

The series for the second term in Machin's formula will decay much faster than the first one. Thus we can restrict our error estimation to the first term.

The last term of the Taylor expansion above had degree 19. Therefore the error is bounded by the following expression: $\max\left|\left(d^{20}(\arctan(y))/(dy^{20})\right)\right|\ (1/5)^{20}\,(1/20!)$, where the maximum is taken over the interval $\left[0,\frac{1}{5}\right]$.

```
d[x_] = D[ArcTan[x], {x, 20}]
```

$$\frac{2\,432\,902\,008\,176\,640\,000\,x}{(x^2+1)^{11}}-\frac{63\,777\,066\,403\,145\,711\,616\,000\,x^{19}}{(x^2+1)^{20}}+\frac{286\,996\,798\,814\,155\,702\,272\,000\,x^{17}}{(x^2+1)^{19}}-\frac{542\,105\,064\,426\,738\,548\,736\,000\,x^{15}}{(x^2+1)^{18}}+\frac{558\,049\,331\,027\,524\,976\,640\,000\,x^{13}}{(x^2+1)^{17}}-\frac{340\,061\,311\,094\,898\,032\,640\,000\,x^{11}}{(x^2+1)^{16}}+\frac{124\,689\,147\,401\,462\,611\,968\,000\,x^{9}}{(x^2+1)^{15}}-\frac{26\,719\,103\,014\,599\,131\,136\,000\,x^{7}}{(x^2+1)^{14}}+\frac{3\,082\,973\,424\,761\,438\,208\,000\,x^{5}}{(x^2+1)^{13}}-\frac{160\,571\,532\,539\,658\,240\,000\,x^{3}}{(x^2+1)^{12}}$$

```
Plot[Abs[d[y] / 20!], {y, 0, 1 / 5}]
```



We see that the maximum of the function on the interval $\left[0, \frac{1}{5}\right]$ is bounded by 0.05. Thus the error for arctan$\left(\frac{1}{5}\right)$ is bounded by:

```
err = 0.05 (1/5)^20
```

5.24288×10^{-16}

Neglecting the much smaller error stemming from the $\frac{1}{239}$-term, we expect our error in approximating π to be bounded by

```
4 err
```

2.09715×10^{-15}

Problem 2:
Perform an error analysis for the Taylor expansion you are using in Problem 1.

Derivation of Machin's Formula

The formula is based on the addition formula for the tangent function:

$$\tan(\alpha + \beta) = \frac{\tan(\alpha) + \tan\beta)}{1 - \tan(\alpha)\tan(\beta)}$$

From this we obtain a formula for $\tan(4\alpha)$:

```
Expand[Numerator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4] /
  Expand[Denominator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4]
```

$$\frac{4\tan(\alpha) - 4\tan^3(\alpha)}{\tan^4(\alpha) - 6\tan^2(\alpha) + 1}$$

Let' s set x = tan (α):

```
(Expand[Numerator[Together[TrigExpand[Tan[4 α]]]] / Cos[α]^4] /
   Expand[Denominator[Together[TrigExpand[Tan[4 α]]]] /
     Cos[α]^4]) /. Tan[α] → x
```

$$\frac{4x - 4x^3}{x^4 - 6x^2 + 1}$$

If we set y = tan (β), and apply the addition formula for the tangent again, we get an expression for tan ($4\alpha + \beta$):

```
Simplify[((4 x - 4 x^3)/(x^4 - 6 x^2 + 1) + y) / (1 - (4 x - 4 x^3)/(x^4 - 6 x^2 + 1) y)]
```

$$\frac{x^4 y - 4x^3 - 6x^2 y + 4x + y}{x^4 + 4x^3 y - 6x^2 - 4xy + 1}$$

We want to use this for $\frac{\pi}{4} = 4\alpha + \beta$, with $\tan\left(\frac{\pi}{4}\right) = 1$:

```
eqn = 1 == Simplify[((4 x - 4 x^3)/(x^4 - 6 x^2 + 1) + y) / (1 - (4 x - 4 x^3)/(x^4 - 6 x^2 + 1) y)]
```

$$1 = \frac{x^4 y - 4x^3 - 6x^2 y + 4x + y}{x^4 + 4x^3 y - 6x^2 - 4xy + 1}$$

We set $x = \frac{1}{5}$ and solve for y:

```
eqn /. {x → 1 / 5}
```

$$1 = \frac{\frac{476 y}{625} + \frac{96}{125}}{\frac{476}{625} - \frac{96 y}{125}}$$

```
Solve[eqn /. {x → 1 / 5}, y]
```

$$\left\{\left\{y \to -\frac{1}{239}\right\}\right\}$$

Consequently $\frac{\pi}{4} = 4 \arctan\left(\frac{1}{5}\right) - \arctan\left(\frac{1}{239}\right)$.

Other Machin-type Formulas

Problem 3:

The critical choice Machin made was to set $x = \frac{1}{5}$. He then obtained $y = \frac{1}{239}$ automatically. Why do you think he considered this to be a good choice for x and y?

(It may be helpful to create a numerical table for the first few odd powers of $\frac{1}{5}$ and $\frac{1}{239}$. A hint for why $y = \frac{1}{239}$ seems to be good choice was already given in the section on Taylor estimates. The next problem may give also you a hint why Machin considered $x = \frac{1}{5}$ to be the optimal choice.)

Problem 4:

What happens when you set $x = \frac{1}{10}$? Why does the resulting y make for a much less suitable "Machin-type" formula?

Extra-credit Problem 5:

Taylor approximations work best when the approximation happens close to the center of the Taylor series. It therefore seems natural to make both x and y as small as possible.

a.) Find the x and y such that the maximum of the two is as small as possible.

b.) Use this x-y- combination and a 10-term Taylor expansion to approximate π. Do you get better results than in the top section of this notebook?

c.) Why did Machin not choose this formula to approximate π?

- **Solution**